
6月 18日の授業中に作成したスケッチ 
//その1 
boolean isOverlapped(float x0, float y0, float r0,  
                     float x1, float y1, float r1) { 
  float d = dist(x0, y0, x1, y1); 
  return d <= (r0+r1); 
  /* 
  if (d <= r0+r1) { 
    return true; 
  } else { 
    return false; 
  } 
  */ 
} 
 
void setup() { 
  size(400, 400); 
} 
 
void draw() { 
  float big = 100; 
  float small = 30; 
 
  background(255); 
  fill(255, 15, 15, 255); 
  ellipse(width/2, height/2, 2*big, 2*big); 
  /* 
  if (!isOverlapped(width/2, height/2, big,  
                    mouseX, mouseY, small)) { 
                      */ 
  if (isOverlapped(width/2, height/2, big,  
                    mouseX, mouseY, small) == false) {                     
    fill(255, 15, 15, 30); 
  } 
  ellipse(mouseX, mouseY, 2*small, 2*small); 
} 
 
//その2 
void setup(){ 
  size(400,400); 
  background(255); 
} 
 
void draw(){ 
} 
 
void mouseDragged(){ 
  fill(128); 
  stroke(128); 
  ellipse(mouseX,mouseY,20,20); 
} 
 
//その3 
float xCircle; 

float yCircle; 
float rCircle; 
boolean drawing; 
 
void setup(){ 
  size(400,400); 
  xCircle = -1; 
  yCircle = -1; 
  rCircle = 0; 
  drawing = false; 
} 
 
void draw(){ 
  background(255); 
  if(drawing){ 
    noFill(); 
    stroke(10,170,255); 
    ellipse(xCircle,yCircle,2*rCircle,2*rCircle); 
    rCircle += 1; 
    if(rCircle > 2*max(width,height)){ 
      drawing = false; 
    } 
  } 
} 
 
void mouseDragged(){ 
  fill(128); 
  stroke(128); 
  ellipse(mouseX,mouseY,20,20); 
} 
 
void mouseReleased(){ 
  drawing = true; 
  xCircle = mouseX; 
  yCircle = mouseY; 
  rCircle = 1; 
} 
 
//その4（動かない） 
void setup(){ 
  size(400,400); 
} 
 
void draw(){ 
  background(255); 
  fill(128); 
  int x = 0; 
  while(x < width){ 
    x += int(random(10)); 
    ellipse(x,height/2,10,10); 
  } 
} 
 
void keyPressed(){ 

  println("pressed"); //<>// 
} 
 
//その5 
boolean isOverlapped(float xTip,float yTip, 
                     float xCircle,float yCircle,float r){ 
  if((xTip-r <= xCircle && xCircle <= xTip+r) 
     && (0 <= yCircle && yCircle <= yTip)){ 
       return true; 
  }else if(dist(xTip,yTip,xCircle,yCircle) <= r){ 
    return true; 
  }else if(dist(xTip,0,xCircle,yCircle) <= r){ 
    return true; 
  } 
  /* 
  else{ 
    return false; 
  } 
  */ 
  return false; 
} 
 
void setup(){ 
  size(400,400); 
} 
 
void draw(){ 
  background(255); 
  stroke(0); 
  float xTip = width/2; 
  float yTip = height/2; 
  line(xTip,0,xTip,yTip); 
  float radius = 20; 
  if(isOverlapped(xTip,yTip,mouseX,mouseY,radius)){ 
    fill(255,10,10); 
  }else{ 
    fill(10,255,10); 
  } 
  ellipse(mouseX,mouseY,2*radius,2*radius); 
} 
 
//その6 
boolean isOverlapped(float xTip,float yTip, 
                     float xCircle,float yCircle,float r){ 
  if((xTip-r <= xCircle && xCircle <= xTip+r) 
     && (0 <= yCircle && yCircle <= yTip)){ 
       return true; 
  }else if(dist(xTip,yTip,xCircle,yCircle) <= r){ 
    return true; 
  }else if(dist(xTip,0,xCircle,yCircle) <= r){ 
    return true; 
  } 
  /* 
  else{ 

    return false; 
  } 
  */ 
  return false; 
} 
 
void setup(){ 
  size(400,400); 
} 
 
void draw(){ 
  background(255); 
  stroke(0); 
  float xTip0 = width/3; 
  float yTip0 = height/3; 
  float xTip1 = 2*width/3; 
  float yTip1 = 2*height/3;   
  line(xTip0,0,xTip0,yTip0); 
  line(xTip1,0,xTip1,yTip1); 
  float radius = 20; 
  if(isOverlapped(xTip0,yTip0,mouseX,mouseY,radius) || 
     isOverlapped(xTip1,yTip1,mouseX,mouseY,radius)){ 
    fill(255,10,10); 
  }else{ 
    fill(10,255,10); 
  } 
  ellipse(mouseX,mouseY,2*radius,2*radius); 
} 
 
//その7 
void mouseWheel(MouseEvent event) { 
  float e = event.getCount(); 
  println(e); 
} 
 
void mousePressed() { 
  println("mouse pressed "+mouseButton); 
} 
 
void mouseDragged() { 
  println("mouse dragged"); 
} 
 
void mouseReleased() { 
  println("mouse released"); 
} 
void mouseMoved() { 
  println("mouse moved"); 
} 
void keyPressed() { 
  println("pressed " + int(key) + " " + keyCode); 
} 
void keyTyped() { 
  println("typed " + int(key) + " " + keyCode); 

} 
void keyReleased() { 
  println("released " + int(key) + " " + keyCode); 
} 
 
void setup() { 
  size(200, 200); 
} 
 
void draw() { 
  background(255); 
} 
 
//その8 
float xPack; 
float yPack; 
 
void drawPackMan() { 
  pushMatrix(); 
  fill(255, 255, 10); 
  float angle=PI/6; 
  arc(0, 0, 40, 40, angle, 2*PI-angle); 
  rotate(angle); 
  line(0, 0, 20, 0); 
  rotate(-2*angle); 
  line(0, 0, 20, 0); 
  popMatrix(); 
} 
 
void setup() { 
  size(400, 400); 
  xPack = width/2; 
  yPack = height/2; 
} 
 
void draw() { 
  background(255); 
  translate(xPack, yPack); 
  drawPackMan(); 
} 
 
void keyPressed() { 
  if (key == CODED) { 
    if (keyCode == LEFT) { 
      xPack -= 1; 
    } else if (keyCode == RIGHT) { 
      xPack += 1; 
    } else if (keyCode == UP) { 
      yPack -= 1; 
    }else if (keyCode == DOWN) { 
      yPack += 1; 
    } 
  }else if(key == 's'){ 
    save("pack"+frameCount+".png"); 

  } 
} 
 


